Rules for integrands of the form $(f+g x)^m (h+i x)^q \left(A+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p$

1. $\int (f+g x)^m (h+i x)^q \left(A+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx$ when $b c-a d\neq 0 \land b f-a g==0 \land d h-c i==0$

1: $\int (f+g x)^m (h+i x) \left(A+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right) dx$ when $b c-a d\neq 0 \land b f-a g==0 \land d h-c i==0 \land m+2\in \mathbb{Z}^+$

Rule: If $b c-a d\neq 0 \land b f-a g==0 \land d h-c i==0 \land m+2\in \mathbb{Z}^+$, then

$$\int (f+g x)^m (h+i x) \left(A+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right) dx \to$$

$$\frac{(f+g x)^{m+1} (h+i x) \left(A+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)}{g (m+2)}+\frac{i (b c-a d)}{b d (m+2)} \int (f+g x)^m \left(A-B n+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right) dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.]),x_Symbol] :=
  (f+g*x)^(m+1)*(h+i*x)*(A+B*Log[e*((a+b*x)/(c+d*x))^n])/(g*(m+2)) +
  i*(b*c-a*d)/(b*d*(m+2))*Int[(f+g*x)^m*(A-B*n+B*Log[e*((a+b*x)/(c+d*x))^n]),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n},x] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && IGtQ[m,-2]
```

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_]),x_Symbol] :=
  (f+g*x)^(m+1)*(h+i*x)*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])/(g*(m+2)) +
  i*(b*c-a*d)/(b*d*(m+2))*Int[(f+g*x)^m*(A-B*n+B*Log[e*(a+b*x)^n/(c+d*x)^n]),x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && IGtQ[m,-2]
```

2: $\int (f+g x)^m (h+i x)^q \left(A+B \operatorname{Log}\left[e \left(\frac{a+b x}{c+d x}\right)^n\right]\right)^p dx$ when $b c-a d\neq 0 \land b f-a g==0 \land d h-c i==0 \land (m \mid q)\in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $F\left[x,\frac{a+b x}{c+d x}\right]==(b c-a d) \operatorname{Subst}\left[\frac{F\left[-\frac{a-c x}{b-d x},x\right]}{(b-d x)^2},x,\frac{a+b x}{c+d x}\right] \partial_x \frac{a+b x}{c+d x}$

Rule: If $b c-a d\neq 0 \land b f-a g==0 \land d h-c i==0 \land (m \mid q)\in \mathbb{Z}$, then

$$\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx\;\to$$
$$(b\,c-a\,d)^{m+q+1}\,\left(\frac{g}{b}\right)^m\,\left(\frac{i}{d}\right)^q\,\text{Subst}\left[\int\frac{x^m\,\left(A+B\,\text{Log}\left[e\,x^n\right]\right)^p}{(b-d\,x)^{m+q+2}}\,dx,\;x,\;\frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (b*c-a*d)^(m+q+1)*(g/b)^m*(i/d)^q*Subst[Int[x^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+q+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,n,p},x] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && IntegersQ[m,q]
```

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (b*c-a*d)^(m+q+1)*(g/b)^m*(i/d)^q*Subst[Int[x^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+q+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,n,p},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && IntegersQ[m,q]
```

3: $\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,b\,f-a\,g==0\,\wedge\,d\,h-c\,i==0\,\wedge\,m+q+2==0$

Derivation: Integration by substitution and partial fraction expansion

Basis: $F\left[x,\frac{a+b\,x}{c+d\,x}\right]==(b\,c-a\,d)\;\text{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},\;x,\;\frac{a+b\,x}{c+d\,x}\right]\,\partial_x\frac{a+b\,x}{c+d\,x}$

Basis: If $m+q+2==0$, then $\partial_x\frac{\left(\frac{g\,(b\,c-a\,d)\,x}{b\,(b-d\,x)}\right)^m\,\left(\frac{i\,(b\,c-a\,d)}{d\,(b-d\,x)}\right)^q}{x^m\,(b-d\,x)^2}==0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,b\,f-a\,g==0\,\wedge\,d\,h-c\,i==0\,\wedge\,m+q+2==0$, then

$$\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx$$

$$\to\;(b\,c-a\,d)\;\text{Subst}\left[\int\frac{\left(\frac{g\,(b\,c-a\,d)\,x}{b\,(b-d\,x)}\right)^m\,\left(\frac{i\,(b\,c-a\,d)}{d\,(b-d\,x)}\right)^q\,\left(A+B\,\text{Log}\left[e\,x^n\right]\right)^p}{(b-d\,x)^2}\,dx,\;x,\;\frac{a+b\,x}{c+d\,x}\right]$$

$$\to\;(b\,c-a\,d)\;\text{Subst}\left[\frac{\left(\frac{g\,(b\,c-a\,d)\,x}{b\,(b-d\,x)}\right)^m\,\left(\frac{i\,(b\,c-a\,d)}{d\,(b-d\,x)}\right)^q}{x^m\,(b-d\,x)^2}\int x^m\,\left(A+B\,\text{Log}\left[e\,x^n\right]\right)^p\,dx,\;x,\;\frac{a+b\,x}{c+d\,x}\right]$$

$$\rightarrow \frac{d^2 \left(\frac{g (a+b x)}{b}\right)^m}{i^2 (b c - a d) \left(\frac{i (c+d x)}{d}\right)^m \left(\frac{a+b x}{c+d x}\right)^m} \text{Subst}\left[\int x^m (A + B \text{Log}[e x^n])^p \, dx, x, \frac{a + b x}{c + d x}\right]$$

–

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  d^2*(g*(a+b*x)/b)^m/(i^2*(b*c-a*d)*(i*(c+d*x)/d)^m*((a+b*x)/(c+d*x))^m)*
    Subst[Int[x^m*(A+B*Log[e*x^n])^p,x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n,p,q},x] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && EqQ[m+q+2,0]
```

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  d^2*(g*(a+b*x)/b)^m/(i^2*(b*c-a*d)*(i*(c+d*x)/d)^m*((a+b*x)/(c+d*x))^m)*
    Subst[Int[x^m*(A+B*Log[e*x^n])^p,x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n,p,q},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && EqQ[m+q+2,0]
```

```
(* Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  b*d*(f+g*x)^(m+1)/(g*i*(b*c-a*d)*(h+i*x)^(m+1)*((a+b*x)/(c+d*x))^(m+1))*
    Subst[Int[x^m*(A+B*Log[e*x^n])^p,x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n,p,q},x] && EqQ[n+mn,0] && NeQ[b*c-a*d,0] && EqQ[b*f-a*g,0] && EqQ[d*h-c*i,0] && EqQ[m+q+2,0] *)
```

2: $\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}^+\wedge\,d\,h-c\,i==0$

Derivation: Integration by substitution

Basis: $F\left[x,\frac{a+b\,x}{c+d\,x}\right]==(b\,c-a\,d)\,\text{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},x,\frac{a+b\,x}{c+d\,x}\right]\partial_x\frac{a+b\,x}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}^+\wedge\,d\,h-c\,i==0$, then

$$\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx\,\rightarrow$$
$$(b\,c-a\,d)^{q+1}\left(\frac{i}{d}\right)^q\text{Subst}\left[\int\frac{(b\,f-a\,g-(d\,f-c\,g)\,x)^m\,(A+B\,\text{Log}[e\,x^n])^p}{(b-d\,x)^{m+q+2}}\,dx,x,\frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (b*c-a*d)^(q+1)*(i/d)^q*Subst[Int[(b*f-a*g-(d*f-c*g)*x)^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+q+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,n},x] && NeQ[b*c-a*d,0] && IntegersQ[m,q] && IGtQ[p,0] && EqQ[d*h-c*i,0]
```

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (b*c-a*d)^(q+1)*(i/d)^q*Subst[Int[(b*f-a*g-(d*f-c*g)*x)^m*(A+B*Log[e*x^n])^p/(b-d*x)^(m+q+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,n},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && IntegersQ[m,q] && IGtQ[p,0] && EqQ[d*h-c*i,0]
```

3: $\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F\left[x,\frac{a+b\,x}{c+d\,x}\right]==(b\,c-a\,d)\,\text{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},x,\frac{a+b\,x}{c+d\,x}\right]\partial_x\frac{a+b\,x}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,(m\mid q)\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}^+$, then

$$\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx\,\longrightarrow$$
$$(b\,c-a\,d)\,\text{Subst}\left[\int\frac{(b\,f-a\,g-(d\,f-c\,g)\,x)^m\,(b\,h-a\,i-(d\,h-c\,i)\,x)^q\,\left(A+B\,\text{Log}[e\,x^n]\right)^p}{(b-d\,x)^{m+q+2}}\,dx,\,x,\,\frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  (b*c-a*d)*Subst[Int[(b*f-a*g-(d*f-c*g)*x)^m*(b*h-a*i-(d*h-c*i)*x)^q*(A+B*Log[e*x^n])^p/(b-d*x)^(m+q+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,n},x] && NeQ[b*c-a*d,0] && IntegersQ[m,q] && IGtQ[p,0]
```

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  (b*c-a*d)*Subst[Int[(b*f-a*g-(d*f-c*g)*x)^m*(b*h-a*i-(d*h-c*i)*x)^q*(A+B*Log[e*x^n])^p/(b-d*x)^(m+q+2),x],x,(a+b*x)/(c+d*x)] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,n},x] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && IntegersQ[m,q] && IGtQ[p,0]
```

U: $\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx$

Rule:

$$\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx\,\longrightarrow\,\int (f+g\,x)^m\,(h+i\,x)^q\,\left(A+B\,\text{Log}\left[e\,\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  Unintegrable[(f+g*x)^m*(h+i*x)^q*(A+B*Log[e*((a+b*x)/(c+d*x))^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n,p,q},x]
```

```
Int[(f_.+g_.*x_)^m_.*(h_.+i_.*x_)^q_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  Unintegrable[(f+g*x)^m*(h+i*x)^q*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,A,B,m,n,p,q},x] && EqQ[n+mn,0] && IntegerQ[n]
```

N: $\int w^m y^q \left(A + B \operatorname{Log}\left[e \left(\frac{u}{v}\right)^n\right]\right)^p dx$ when $u == a + b x \land v == c + d x \land w == f + g x \land y == h + i x$

Derivation: Algebraic normalization

Rule: If $u == a + b x \land v == c + d x \land w == f + g x \land y == h + i x$, then

$$\int w^m y^q \left(A + B \operatorname{Log}\left[e \left(\frac{u}{v}\right)^n\right]\right)^p dx \longrightarrow \int (f + g x)^m (h + i x)^q \left(A + B \operatorname{Log}\left[e \left(\frac{a + b x}{c + d x}\right)^n\right]\right)^p dx$$

Program code:

```
Int[w_^m_.*y_^q_.*(A_.+B_.*Log[e_.*(u_/v_)^n_.])^p_.,x_Symbol] :=
  Int[ExpandToSum[w,x]^m*ExpandToSum[y,x]^q*(A+B*Log[e*(ExpandToSum[u,x]/ExpandToSum[v,x])^n])^p,x] /;
FreeQ[{e,A,B,m,n,p,q},x] && LinearQ[{u,v,w,y},x] && Not[LinearMatchQ[{u,v,w,y},x]]
```

```
Int[w_^m_.*y_^q_.*(A_.+B_.*Log[e_.*u_^n_.*v_^mn_])^p_.,x_Symbol] :=
  Int[ExpandToSum[w,x]^m*ExpandToSum[y,x]^q*(A+B*Log[e*ExpandToSum[u,x]^n/ExpandToSum[v,x]^n])^p,x] /;
FreeQ[{e,A,B,m,n,p,q},x] && EqQ[n+mn,0] && IGtQ[n,0] && LinearQ[{u,v,w,y},x] && Not[LinearMatchQ[{u,v,w,y},x]]
```

S: $\int w \left(A + B \operatorname{Log}\left[e \frac{u^n}{v^n}\right]\right)^p dx$ when $u == a + b x \land v == c + d x \land n \notin \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\partial_x \operatorname{Log}\left[e \frac{u[x]^n}{v[x]^n}\right] == \partial_x \operatorname{Log}\left[e \left(\frac{u[x]}{v[x]}\right)^n\right]$

Rule: If $u == a + b x \land v == c + d x \land n \notin \mathbb{Z}$, then

$$\int w \left(A + B \operatorname{Log}\left[e \frac{u^n}{v^n}\right]\right)^p dx \longrightarrow \operatorname{Subst}\left[\int w \left(A + B \operatorname{Log}\left[e \left(\frac{u}{v}\right)^n\right]\right)^p dx,\ e \left(\frac{u}{v}\right)^n,\ e \frac{u^n}{v^n}\right]$$

Program code:

```
Int[w_.*(A_.+B_.*Log[e_.*u_^n_.*v_^mn_])^p_.,x_Symbol] :=
  Subst[Int[w*(A+B*Log[e*(u/v)^n])^p,x],e*(u/v)^n,e*u^n/v^n] /;
FreeQ[{e,A,B,n,p},x] && EqQ[n+mn,0] && LinearQ[{u,v},x] && Not[IntegerQ[n]]
```

```
(* Int[w_.*(A_.+B_.*Log[e_.*(f_.*u_^q_.*v_^mq_)^n_.])^p_.,x_Symbol] :=
  Subst[Int[w*(A+B*Log[e*f^n*(u/v)^(n*q)])^p,x],e*f^n*(u/v)^(n*q),e*(f*(u^q/v^q))^n] /;
FreeQ[{e,f,A,B,n,p,q},x] && EqQ[q+mq,0] && LinearQ[{u,v},x] && Not[IntegerQ[n]] *)
```

Rules for integrands of the form $\left(f+g\,x+h\,x^2\right)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p$

1: $\int \left(f+g\,x+h\,x^2\right)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$ when $b\,d\,f-a\,c\,h==0 \land b\,d\,g-h\,(b\,c+a\,d)==0 \land m\in\mathbb{Z}$

– Derivation: Algebraic simplification

Basis: If $b\,d\,f-a\,c\,h==0 \land b\,d\,g-h\,(b\,c+a\,d)==0$, then $f+g\,x+h\,x^2 == \frac{h}{b\,d}\,(a+b\,x)\,(c+d\,x)$

– Rule: If $b\,d\,f-a\,c\,h==0 \land b\,d\,g-h\,(b\,c+a\,d)==0 \land m\in\mathbb{Z}$, then

$$\int \left(f+g\,x+h\,x^2\right)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx \to \frac{h^m}{b^m\,d^m}\int (a+b\,x)^m\,(c+d\,x)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$$

Program code:

```
Int[(f_.+g_.*x_+h_.*x_^2)^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  h^m/(b^m*d^m)*Int[(a+b*x)^m*(c+d*x)^m*(A+B*Log[e*((a+b*x)/(c+d*x))^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,A,B,n,p},x] && EqQ[b*d*f-a*c*h,0] && EqQ[b*d*g-h*(b*c+a*d),0] && IntegerQ[m]
```

```
Int[(f_.+g_.*x_+h_.*x_^2)^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  h^m/(b^m*d^m)*Int[(a+b*x)^m*(c+d*x)^m*(A+B*Log[e*(a+b*x)^n/(c+d*x)^n])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,h,A,B,n,p},x] && EqQ[n+mn,0] && IGtQ[n,0] && EqQ[b*d*f-a*c*h,0] && EqQ[b*d*g-h*(b*c+a*d),0] && IntegerQ[m]
```

2: $\int \left(f+g\,x+h\,x^2\right)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F\left[x,\,\frac{a+b\,x}{c+d\,x}\right]=(b\,c-a\,d)\,\text{Subst}\left[\frac{F\left[-\frac{a-c\,x}{b-d\,x},x\right]}{(b-d\,x)^2},\,x,\,\frac{a+b\,x}{c+d\,x}\right]\partial_x\frac{a+b\,x}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}^+$, then

$$\int \left(f+g\,x+h\,x^2\right)^m \left(A+B\,\text{Log}\left[e\left(\frac{a+b\,x}{c+d\,x}\right)^n\right]\right)^p dx\,\longrightarrow$$

$$(b\,c-a\,d)\,\text{Subst}\left[\int\frac{1}{(b-d\,x)^{2\,(m+1)}}\left(b^2\,f-a\,b\,g+a^2\,h-\left(2\,b\,d\,f-b\,c\,g-a\,d\,g+2\,a\,c\,h\right)\,x+\left(d^2\,f-c\,d\,g+c^2\,h\right)\,x^2\right)^m \left(A+B\,\text{Log}\left[e\,x^n\right]\right)^p dx,\,x,\,\frac{a+b\,x}{c+d\,x}\right]$$

Program code:

```
Int[P2x_^m_.*(A_.+B_.*Log[e_.*((a_.+b_.*x_)/(c_.+d_.*x_))^n_.])^p_.,x_Symbol] :=
  With[{f=Coeff[P2x,x,0],g=Coeff[P2x,x,1],h=Coeff[P2x,x,2]},
  (b*c-a*d)*
    Subst[Int[(b^2*f-a*b*g+a^2*h-(2*b*d*f-b*c*g-a*d*g+2*a*c*h)*x+(d^2*f-c*d*g+c^2*h)*x^2)^m*(A+B*Log[e*x^n])^p/
      (b-d*x)^(2*(m+1)),x],x,(a+b*x)/(c+d*x)]] /;
FreeQ[{a,b,c,d,e,A,B,n},x] && PolyQ[P2x,x,2] && NeQ[b*c-a*d,0] && IntegerQ[m] && IGtQ[p,0]
```

```
Int[P2x_^m_.*(A_.+B_.*Log[e_.*(a_.+b_.*x_)^n_.*(c_.+d_.*x_)^mn_])^p_.,x_Symbol] :=
  With[{f=Coeff[P2x,x,0],g=Coeff[P2x,x,1],h=Coeff[P2x,x,2]},
  (b*c-a*d)*
    Subst[Int[(b^2*f-a*b*g+a^2*h-(2*b*d*f-b*c*g-a*d*g+2*a*c*h)*x+(d^2*f-c*d*g+c^2*h)*x^2)^m*(A+B*Log[e*x^n])^p/
      (b-d*x)^(2*(m+1)),x],x,(a+b*x)/(c+d*x)]] /;
FreeQ[{a,b,c,d,e,A,B,n},x] && PolyQ[P2x,x,2] && EqQ[n+mn,0] && IGtQ[n,0] && NeQ[b*c-a*d,0] && IntegerQ[m] && IGtQ[p,0]
```